

# File Copier

Martin Engert

*"File Copier" is a BASIC utility that lets you transfer files from one disk to another, using a single drive, without worrying about starting addresses or machine language. For any VIC-20 or Commodore 64.*

**"F**ile Copier" can help those who want to copy sequential or program files from one disk to another but have only a single disk drive and no machine language monitor. Since the program is written in BASIC, it's a bit slow. But one advantage of this program is that you don't have to know the initial address or length of the program to be transferred.

File Copier works on both the VIC-20 and Commodore 64. The program first resets the top-of-BASIC pointers to reserve 1K of memory for itself. The remaining memory is used to store your file temporarily. VIC users should make sure enough memory is available for this purpose before running the program; any amount of expansion memory can be added if necessary.

Each byte of the file is read from disk using the GET# command and POKEd into free memory. Then you insert the new disk and the program writes these bytes onto it using PRINT#. After the file is copied, the top-of-BASIC pointers are restored to normal.

Screen instructions are provided within the program for easier use.

## File Copy for VIC and 64

*For error-free program entry, be sure to use "The Automatic Proofreader," Appendix C.*

```
10 POKE251,PEEK(52) :rem 49
20 POKE52,PEEK(44)+4:POKE56,PEEK(52):CLR :rem 89
30 PRINT"{CLR}RUN THIS PROGRAM TO" :rem 175
40 PRINT"COPY A PROGRAM OR" :rem 106
50 PRINT"SEQUENTIAL FILE FROM" :rem 133
60 PRINT"ONE DISK (THE SOURCE" :rem 30
70 PRINT"DISK) TO ANOTHER (THE" :rem 73
80 PRINT"DESTINATION DISK). " :rem 253
90 PRINT"INSERT SOURCE DISK. " :rem 57
100 M=256*PEEK(52) :rem 191
110 OPEN15,8,15 :rem 32
```

## 5: Utilities and Programming Aids

---

```
120 PRINT"WHAT IS THE NAME OF"           :rem 203
130 PRINT"THE FILE OR PROGRAM":INPUTF$    :rem 83
140 T$="P":PRINT"WHAT IS THE FILE TYPE"   :rem 252
150 PRINT"(P FOR PROGRAM, S FOR"         :rem 68
160 PRINT"FILE)"                          :rem 177
170 INPUTT$                               :rem 160
180 OPEN2,8,2,F$+",""+T$+",R"             :rem 128
190 INPUT#15,E,E$,X,X:IFE<>0THENPRINTE$:CLOSE2:GOT
    O120                                   :rem 134
200 GET#2,A$:IFA$=""THEN A$=CHR$(0)       :rem 90
210 POKEM+J,ASC(A$):J=J+1:IFST=0THEN200   :rem 66
220 CLOSE2                                :rem 60
230 PRINT"INSERT DESTINATION"             :rem 125
240 PRINT"DISK AND PRESS {RVS}RETURN"     :rem 228
250 PRINT"TO COPY."                      :rem 116
260 GETC$:IFC$<>CHR$(13)THEN260          :rem 6
270 PRINT"PRESS {RVS}RETURN{OFF} IF YOU"  :rem 7
280 PRINT"WANT TO KEEP THE NAME"         :rem 111
290 PRINTF$                               :rem 146
300 INPUT"FILE NAME ";F$                  :rem 77
310 OPEN2,8,2,F$+",""+T$+",W"            :rem 128
320 INPUT#15,E,E$,X,X:IFE<>0THENPRINTE$:CLOSE2:GOT
    O300                                   :rem 129
330 FORK=0TOJ-1:PRINT#2,CHR$(PEEK(M+K));:NEXT
                                           :rem 7
340 CLOSE2:CLOSE15                       :rem 85
350 POKE52,PEEK(251):POKE56,PEEK(251):CLR :rem 145
```